
Bayesian chronological modelling 

We have used the Bayesian chronological modelling program OxCal v4 (Bronk Ramsey 

2009) to interpret radiocarbon results, and to estimate the date of initial settlement, at five 

sites in Venice. Here, we provide the essential details to allow these estimates to be 

reproduced or revised. Two models are provided below. Model 1 (shown in Figure 3) is a 

wiggle-match of the Eastern Veneto oak chronology (Martinelli & Kromer 2002) to the 

IntCal13 calibration curve (Reimer et al. 2013), which was not available in 2002. Model 2 

contains the five site sequences, which are fully independent of each other, except that Ca’ 

Foscari, Teatro Malibran and ex-Cinema San Marco are linked by dendrochronological cross-

matching. Part of the Model 2 output is shown in Figure 4. 

Bayesian models combine independent information about the chronology of individual 

samples and events—such as calibrated radiocarbon dates, sapwood estimates, calendar date 

ranges for coins etc. (known as likelihoods)—with observations or suppositions about the 

relative dates of these samples and events (prior information), which may be based on 

stratigraphic sequences, dendrochronological cross-matching, or even a basic assumption that 

their dates are not independent of each other.  

Several programs are available which use computationally intensive algorithms to repeatedly 

sample all the likelihoods in a model, and reject the resulting dates if they are incompatible 

with the prior information provided. This process therefore retains only the potential dates of 

samples and associated events which are consistent with the relative dating constraints 

imposed by the prior information, creating probability distributions known as posterior 

density estimates of the dates of these samples and events. Such estimates are not 

independent of each other, and will change if any of the likelihoods or priors changes, but 

Bayesian modelling is a robust and reproducible method for estimating the real dates of 

individual samples, and of significant events related to those samples.  

 

Likelihoods: radiocarbon data and dendrochronological cross-matching 

The radiocarbon ages and dendrochronological cross-matching positions at Ca’ Foscari, 

Teatro Malibran and ex-Cinema San Marco are identical to those given in Meadows et al. 

(2012), except that the felling dates of dendrochronologically cross-matched timbers are 

expressed as offsets relative to the posterior estimate generated by Model 1 for "year 0" in the 

Eastern Veneto chronology, rather than relative to the last surviving annual ring of Ca’ 

Foscari US1879 (year 199); the 2012 model also used the previous calibration curve. A 

wood-age offset has been applied to the unidentified wood sample from coring at Ca’ Foscari 

in 2002. All the other samples from these sites are from area excavations.  

The Piazza San Marco (St Mark’s Square) model is based on radiocarbon ages of samples 

obtained by coring, reported by Martinelli (2003) and Ammerman et al. (2017: Table S1), 

which incorrectly gives the radiocarbon age of OxA-4521 as 1400±60 BP; the correct result, 

1440±60 BP (Hedges et al. 1995), was given in earlier publications (e.g. Ammerman et al. 

1995; see postscript at the end of this section). The San Lorenzo di Castello radiocarbon  

results (Ammerman et al. 1992; Hedges et al. 1992; Bronk Ramsey et al. 2002) are from 

samples obtained by a combination of area excavation and coring. There is no relevant 

dendrochronological cross-matching at Piazza San Marco or San Lorenzo di Castello.  

 

Wood-age offsets 

Most of the radiocarbon results from these five sites are on samples of wood, which can be 

misleadingly old if the sample consists of tree-rings formed long before the tree-fall or felling 

date. If no correction is made for wood-age offsets, model estimates of the initial dates of 

settlement will be too early.  Almost all samples at Ca’ Foscari, Teatro Malibran and ex-

Cinema San Marco were selected by a dendrochronologist; the number of tree rings in each 

sample, and the interval between the midpoint of the sample and the felling date of the 

timber, is usually known exactly. These offsets can be coded in OxCal; e.g. R_Date("Hd-

24673", 1418, 17)+3; generates a probability density function for the felling date, by 

shifting the probability density function of the calibrated date exactly 3 years later (in this 

case, the sample was complete to the bark edge and contained 6 annual rings). If the wood-

age offset is not known exactly, a likelihood can be specified for the number of missing 

annual rings between the midpoint of the sample and its felling date. In the special case of 

mature oak-wood retaining the heartwood-sapwood transition, a sapwood estimate (Corona 

1974) can be applied (e.g. +U(11,40) would shift the date by a uniform probability 

distribution of 11–40 years).  

Martinelli (2003) reported two series of radiocarbon results on wood from coring at St 

Mark’s Basilica: seven samples dated by AMS (OxA-) and ten radiometric dates from 

Heidelberg (Hd-). The AMS samples were not identified before submission and although it 

can be assumed that only the outermost surviving rings were dated, appropriate wood-age 

offsets may be quite variable, as they depend on the number of missing outer rings. The 

Heidelberg samples were identified, and the number of annual rings present was estimated. 

We assume that the entire sample was processed in these cases. This means that the average 

intrinsic age of the radiometric samples may be larger but less uncertain, as part of the offset 

(half the number of annual rings present) is known. All the Ammerman et al. samples were 

dated by AMS. Some were short-lived (the peach stones, a bone, a reed), while others were 

identified as elm or oak wood, with no information about maturity. 

As most of the identified samples from coring were of oak (Quercus sect. ROBUR) (or less 

commonly elm (Ulmus sp.), or ash (Fraxinus sp.), which are similarly long-lived), we assume 

that most of the unidentified wood was also oak. Typically we have applied offsets of 20–60 

years to such samples, unless the timber was described as a pole (i.e. shorter-lived) or very 

slow-grown (longer-lived). These species can live much longer, as shown by the 

dendrochronological cross-matching, but unless the median intrinsic age of such samples is 

assumed to be more than e.g. 100 years, our model favours a seventh-century date for the 

start of settlement, both at Piazza San Marco and at San Lorenzo di Castello. If the estimated 

wood-age offsets are less precise (e.g. 10–70 rather than 20–60 years), the uncertainty in the 

model output increases. However, even if unrealistically small and precise wood-age offsets 

(e.g. 10–20 years) are applied to wood samples from coring, it is unlikely, based on our 

model, that settlement at either location began much before the mid-seventh century. 

 

Prior information  

The prior information in the Ca’ Foscari, Teatro Malibran and ex-Cinema San Marco models 

is identical to that in Meadows et al. (2012). The Ca’ Foscari model incorporates the 

stratigraphic sequence (the Harris matrix), whereas the ex-Cinema San Marco results are 

placed in simple bounded-phase model, as the relevant stratigraphic relationships are 

unpublished. For Teatro Malibran, only the dendrochronologically cross-matched timbers are 

included, and the only prior information is therefore their cross-matching positions. 

Our model for Piazza San Marco assumes that there was a pre-Basilica settlement phase, 

which ended when construction of the first Basilica began in AD 828 (when, according to 

tradition, the relics of St Mark were brought to Venice). The Ammerman et al. (2017) 

samples are placed in the pre-Basilica phase, while the Martinelli samples are placed after the 

start of the Basilica; their dates are consistent with the original interpretation that they came 

from structural timbers predating the current, eleventh-century, Basilica. Our model gives an 

unrealistically precise estimated date for the start of settlement at Piazza San Marco, because 

it incorporates as prior information the nominal sequences of samples within cores, when 2–3 

samples have been dated per core. In each case, it is possible that the stratigraphically earlier 

sample predates the stratigraphically later sample, but the reverse sequence is usually also 

possible, and all samples may be redeposited. Removing this prior information only increases 

the uncertainty in the estimated start date by 5–10 years, however.    

It was possible to create a less subjective model for the San Lorenzo di Castello sequence, 

using information provided at the time of radiocarbon sample submission, and subsequently 

published in Archaeometry datelists (Hedges et al. 1992; Bronk Ramsey et al. 2002). The 

1992 list is actually headed San Lorenzo di Ammiana, correctly described as being near 

Torcello in the northern Lagoon, but incorrectly located at 45°12’N 12°21’E (in the Adriatic, 

east of Chioggia). The description of the church by ‘AJA’ is consistent with San Lorenzo di 

Castello, however, and with the publication of the same dates in Ammerman et al. (1992), 

where they are clearly attributed to Castello. The second datelist (Bronk Ramsey et al. 2002), 

which repeats the incorrect coordinates, attributes both sets of samples to San Lorenzo di 

Castello.  

Our model assumes that a reed, OxA-3338 (‘from a natural horizon in the lagoonal (barena) 

sediments at a depth of about one metre below the mortar fragments’), predates all the 

anthropogenic samples, while the first settlement phase includes OxA-3339 (‘a reed structure 

observed at the base of the trial sounding’), mortar encountered by coring (‘OxA-3334 and -

3335 bracket a level with mortar fragments in core 13. OxA-3336 to -3337 do the same for a 

similar level in core 13’), and what may be volparoni (‘a dense layer of twigs and canes… 

[which] represents a phase of land reclamation on the site’s west side at a time prior to the 

construction of the first church’ [OxA-7012, 8199] and ‘a small wooden stake driven into a 

canal bank’ [OxA-6769]). There are also dates on three structural timbers and three burials 

associated with the church itself, believed to originate in the early ninth century. In the model 

code, it is assumed that the burials took place after the church was built, and before the latest 

structural timber, timber 8, was felled. Although no palaeodietary information is given, it is 

unlikely that these individuals had fully terrestrial diets if they lived in Venice; their 

radiocarbon ages are therefore calibrated assuming that 25±10 per cent of protein consumed 

was from marine species. However wrong this estimate may be, the burial dates do not affect 

the chronology of the initial settlement phase; in fact, our estimated date of the start of 

settlement at San Lorenzo would not be affected if the church phase was omitted entirely.  

 

Postscript 

After it was published in December 2018, the Antiquity editorial office drew our attention to 

the fact that whilst our article was under review, three of the remarkably early AMS dates 

from coring in St Mark’s Basilica (OxA-9339, 9340 and 9341, which our model treated as 

termini post quos for settlement in the Piazza San Marco) had been withdrawn, and replaced 

by new measurements of the same samples (OxA-33383, 33384 and 33793 respectively) 

(Higham et al. 2018). OxA-33793 was in fact a new measurement of the sample previously 

reported as OxA-9342 (Higham et al. 2020), whereas OxA-9341 had been withdrawn some 

years earlier, and replaced by OxA-11542 (T.F.G. Higham, pers comm). OxA-11542 

(1034±31 BP) was reported by Ammerman et al. (2019), which also includes AA-108917 

(941±31 BP), on a wooden pole from core CF 28 beneath the Basilica (which appears not to 

have been dated previously).  

Our Model 2 (below) has therefore been updated, to include the new results and omit the 

withdrawn measurements. As these new dates are not problematically early, they are now 

included in the phase of structural timbers supporting the Basilica. One pre-seventh century 

date from coring underneath the Basilica (OxA-9516, 1575±60 BP (Martinelli 2003)) appears 

not to have been withdrawn yet (ORAU Database, available at c14.arch.ox.ac.uk, accessed 13 

Aug 2020), contrary to footnote 31 of Ammerman et al. (2019). In view of the technical 

issues affecting many samples first measured in 1999 (Bronk Ramsey et al. 2002), and the 

withdrawal of the other three early dates after re-measurement, we suspect that OxA-9516 

might also be inaccurate, and have not used it in the updated model. The updated model’s 

estimated date of first settlement in the Piazza San Marco are is essentially identical to that 

shown in our 2018 article.  

Model 1. Wiggle-match dating of the Eastern Veneto floating chronology and some 

cross-matched timbers (Figure 3).   

The following OxCal CQL code (equal-area font) will run in OxCal v4 (Bronk Ramsey 

2009).   

Options() 

 { 

  Resolution=1; 

 }; 

 Plot() 

 { 

  //details in Martinelli and Kromer (2002) 

  D_Sequence("Eastern Veneto oak chronology wiggle-match") 

  { 

   Date("year 0"); 

   Gap(51); 

   R_Date("year 51: Hd-21355", 1588, 16); 

   Gap(73); 

   R_Date("year 124: Hd-20714", 1465, 24); 

   Gap(74); 

   R_Date("year 198: Hd-20713", 1439, 23); 

   Gap(52); 

   Date("year 250"); 

  }; 

  Line( ); 

  //cross-matching positions shown in Figure 2 

  //further details in Meadows et al. (2012) 

  //post US1282 ends in year 201 at the heartwood-sapwood 

transition, and is estimated to be missing 11–40 sapwood 

rings, giving a felling date range of years 202–231 

  MCMC_Sample("felling date estimates for cross-matched timber 

structures") 

  { 

   Label("Teatro Malibran, Structure E, year 212");   

Date("=year 0")+212; 

   Label("Ca' Foscari, plank US177, year 226");   Date("=year 

0")+226; 

   Label("ex-Cinema San Marco, post US1282, year 202-231");   

Date("=year 0")+U(202,231); 

  }; 

 }; 

The posterior estimate for "year 0" generated by this model should be saved as year_0.prior 

in the same folder as Model 2, in order for Model 2 to run properly. 

 

Model 2. Site chronologies for five locations (Figure 4).   
Plot() 

 { 

  Prior("year 0", "year_0.prior"); 

  Phase("five early settlements")  

  { 

   Sequence("Ca' Foscari")//model incorporates stratigraphic 

sequence of structures 

   { 

    Boundary("Ca' Foscari start settlement"); 

    Phase("Ca' Foscari samples") 

    { 

     R_Date("2002 sample", 1460, 50)+U(20,60);//unidentified 

wood from coring below limit of excavation 

     Sequence("2004 excavation") 

     { 

      Phase("Wattle US 1660") 

      { 

       R_Date("Hd-24673", 1418, 17)+3; 

      }; 

      Phase("Phase 2 onwards") 

      { 

       Sequence("US 1940 branch") 

       { 

        Phase("Wattle US 1940") 

        { 

         R_Date("Hd-24703", 1341, 17)+4; 

        }; 

        Phase("Wattle US 1814") 

        { 

         R_Date("Hd-24684", 1321, 19)+2; 

        }; 

        Phase("US 1757, 1871") 

        { 

         Phase("Wattle US 1757A") 

         { 

          R_Date("Hd-24704", 1270, 17)+3; 

         }; 

         Phase("Wattle US 1871") 

         { 

          R_Date("Hd-24706", 1297, 18)+2; 

         }; 

        }; 

       }; 

       Sequence("US 1879 and later structures") 

       { 

        //oak heartwood, final ring is year 199; at least 11 

sapwood rings missing 

        After("Plank US 1879") 

        { 

         Date("=year 0")+210; 

        }; 

        //elm complete to bark-edge, final ring is year 226 

        Phase("Plank US 177") 

        { 

         Date("=year 0")+226; 

        }; 

        Phase("US 1882, 2121") 

        { 

         Phase("Wattle US 2121") 

         { 

          R_Date("Hd-24683", 1293, 21)+3; 

         }; 

         Sequence("US 1882 onwards") 

         { 

          Phase("Wattle US 1882") 

          { 

           R_Date("Hd-24688", 1283, 19)+3; 

          }; 

          Phase("Wattle US 2000") 

          { 

           R_Date("Hd-24687", 1278, 21)+2; 

          }; 

         }; 

        }; 

       }; 

      }; 

     }; 

    }; 

    Boundary("Ca' Foscari end phase 3"); 

   }; 

   Line( ); 

   Phase("Teatro Malibran") 

   { 

    //multiple oak timbers complete to bark-edge, final ring 

is year 212 

    Phase("structure E") 

    { 

     Date("=year 0")+212; 

    }; 

    //elm, no bark-edge, no sapwood estimate, last cross-

matched ring is year 227 

    After("structure C") 

    { 

     Date("=year 0")+227; 

    }; 

    First("Teatro Malibran Structure E"); 

   }; 

   Line( ); 

   Sequence("ex-Cinema San Marco")//simple bounded-phase 

model, as detailed stratigraphic sequence is not yet available 

   { 

    Boundary("ex-Cinema San Marco start settlement"); 

    Phase("ex-Cinema San Marco samples") 

    { 

     //oak, heartwood only, final ring is year 134, minimum 11 

sapwood rings missing 

     After ("US 1317") 

     { 

      Date("=year 0")+145; 

     }; 

     //oak, complete to heartwood/sapwood boundary at year 

191, sapwood estimate 11–40 rings 

     Phase ("US 1282 post") 

     { 

      Date("=year 0")+U(202, 231); 

     }; 

     //complete to bark-edge in year 241 

     Phase("US 1211 post") 

     { 

      Date("=year 0")+241; 

     }; 

     //one timber is complete to bark-edge at year 236 

     Phase("US 1212/29") 

     { 

      Date("=year 0")+236; 

      R_Date("Hd-22405", 1281, 19); 

     }; 

     Phase("US 1223/8") 

     { 

      R_Date("Hd-22399", 1267, 15); 

     }; 

     Phase("US 1275/3") 

     { 

      R_Date("Hd-22073", 1302, 12); 

     }; 

     Phase("EUS 1330") 

     { 

      R_Date("Hd-21983", 1305, 20); 

     }; 

    }; 

    Boundary("ex-Cinema San Marco end"); 

   }; 

   Line( ); 

   Sequence("Piazza San Marco") 

   { 

    Boundary("Piazza San Marco start settlement"); 

    Phase("Piazza San Marco ") 

    { 

     Phase("pre-Basilica occupation") 

     { 

      R_Date("OxA-4521 ash wood", 1440, 60)+U(20,60); 

      Sequence("core 3 Marciana") 

      { 

       R_Date("OxA-4525 reed", 1305, 50); 

       R_Date("OxA-4524 oak wood", 1295, 55)+U(20,60); 

       Phase("core 3 -3.1m") 

       { 

        R_Date("OxA-4522 reed", 1340, 50); 

        R_Date("OxA-4523 elm wood", 1260, 55)+U(20,60); 

       }; 

      }; 

      Sequence("core BSM4 Basilica") 

      { 

       Phase("peach stones") 

       { 



12 

        R_Date("AA-107037 peach stone", 1311, 25); 

        R_Date("AA-107038 peach stone", 1310, 21); 

       }; 

       R_Date("AA-107039 elm wood", 1326, 21)+U(20,60); 

      }; 

      Sequence("core 56 Piazzetta") 

      { 

       R_Date("OxA-5208 wood chip", 1325, 55)+U(10,20); 

       R_Date("OxA-5323 bone", 1325, 40); 

      }; 

      Sequence("core B5 Piazza") 

      { 

       R_Date("OxA-5209 oak wood", 1300, 55)+U(20,60); 

       R_Date("replicated elm wood (OxA-5324, x28622)", 1307, 

31)+U(20,60); 

      }; 

     }; 

    }; 

    Boundary("start building Basilica",828); 

    Phase("structural timbers") 

    { 

     R_Date("Hd-21469 CF 1 oak wood", 1108, 17)+U(20,40); 

     R_Date("Hd-22117 CF 32 oak wood", 1085, 14)+U(30,60); 

     R_Date("Hd-21893 CF 61 oak wood", 1070, 23)+U(20,40); 

     R_Date("Hd-21595 CF 43 oak wood", 956, 27)+U(25,50); 

     R_Date("Hd-21459 CF 38 alder wood", 905, 19)+U(10,30); 

     R_Date("Hd-21454 CF 34 oak wood", 1176, 17)+U(25,50); 

     R_Date("OxA-10076 CF 42 wood", 1045, 34)+U(20,60); 

     R_Date("Hd-22119 CF 23 oak wood", 1036, 14)+U(25,50); 

     R_Date("Hd-22120 CF 33 oak wood", 1126, 13)+U(20,40); 

     R_Date("Hd-21467 CF 41 oak wood", 1104, 19)+U(20,40); 

     R_Date("Hd-22112 CF 22 oak wood", 1097, 16)+U(20,40); 

     R_Date("OxA-10077 CF 19 wood", 965, 36)+U(20,60); 

     R_Date("OxA-11542 CF 42 elm wood", 1034, 31)+U(20,60); 

     R_Date("OxA-33383 CF 2 wood", 1152, 30)+U(20,60); 

     R_Date("OxA-33384 CF 11 wood", 1088, 28)+U(20,60); 

     R_Date("OxA-33793 CF 44 wood", 1172, 27)+U(20,60); 

     R_Date("AA-108917 wood", 941, 31)+U(20,40); 

    }; 

   }; 

   Line( ); 

   Sequence("San Lorenzo di Castello") 

   { 

    R_Date("OxA-3338 6/9-16,-3.20m reed", 1340, 70); 

    Boundary("San Lorenzo di Castello start settlement"); 

    Phase("before church") 

    { 

     R_Date("OxA-3339 17 reed structure", 1380, 55); 

     Sequence("core 13") 

     { 

      R_Date("OxA-3335 3/13-9b,-2.79m wood", 1415, 

65)+U(20,60); 

      Date("mortar in core 13"); 

      R_Date("OxA-3334 2/13-7,-2.51m wood", 1440, 

65)+U(20,60); 

     }; 

     Sequence("core 9") 

     { 

      R_Date("OxA-3337 5/9-13,-2.39m wood", 1390, 

70)+U(20,60); 

      Date("mortar in core 9"); 

      R_Date("OxA-3336 4/9-10,-2.12m wood", 1350, 

70)+U(20,60); 

     }; 

     R_Date("US1764 twigs (OxA-7012, 8199)", 1283, 33); 

     R_Date("OxA-6769 No.10 wood", 1220, 50)+U(5,20); 

    }; 

    Boundary("end early settlement"); 

    Combine("San Lorenzo first church") 

    { 

     R_Date("OxA-3333 1/Nr.13,-1.28m wood", 1155, 

65)+U(10,20); 

     R_Date("OxA-6768 No.9 wood", 1295, 55)+U(20,60); 

    }; 

    Phase("burials")  

    { 

     Curve("IntCal13","IntCal13.14c"); 

     Curve("Marine13","Marine13.14c"); 

     Mix_Curve("Mixed","IntCal13","Marine13",25,10); 

     R_Date("OxA-6747 No.12 bone", 1210, 50); 

     R_Date("OxA-6661 No.13 bone", 1135, 45); 

     R_Date("OxA-6557 No.11 bone", 1115, 45); 

    }; 

    R_Combine("timber 8 OxA-7011, OxA-8198, Hd-17754")//the 3 

14 ages from this structural timber are combined before 

calibration  

    { 

     R_Date("OxA-8198 No.8 wood", 1075, 35); 

     R_Date("OxA-7011 No.8 wood", 1020, 70); 

     R_Date("Hd-17754 outer rings", 1079, 26); 

    }; 

    Boundary("rebuild church"); 

   }; 

  }; 

  Page( );//Figure 4 

  Phase("estimated start of settlement at different sites") 

  { 

   Date("=Ca' Foscari start settlement"); 

   Date("=Teatro Malibran Structure E"); 

   Date("=ex-Cinema San Marco start settlement"); 

   Date("=Piazza San Marco start settlement"); 

   Date("=San Lorenzo di Castello start settlement"); 

  }; 

 }; 

